
COS 335
DISPTIME - Assembler Program to Display the Time

; DISPTIME.A86 -- Version 1
; Obtain system time from DOS and display it on the console

 JMP START
   hstr DB 2 DUP ?,':'   ; string to store hours,
   mstr DB 2 DUP ?,':'   ; minutes
   sstr DB 2 DUP ?,'.'   ; seconds 
   dstr DB 2 DUP ?,'$'   ; and hundredths of seconds
 
   hrs  DB ?             ; binary storage for results
   mins DB ?             ; of DOS Get Time function
   secs DB ?
   hund DB ?
 
 start:
 mov ah,2ch               ; get system time
 int 21h
 mov hrs,ch               ; store the results
 mov mins,cl
 mov secs,dh
 mov hund,dl
 
 call timestr             ; convert time to a string
 mov dx,offset hstr       ; display on the console
 mov ah,09h
 int 21h
 mov ax,4c00h             ; DOS Terminate program
 int 21h      
 
 timestr:
   sub ah,ah               ; zero out ah for future calls
   mov bx, OFFSET hstr+1   ; setup for bin2dec, bx points to end of buffer
   mov cx,2                ; number of digits
   mov al, hrs             ; number to convert
   call bin2dec            ; convert it
 
   mov bx, OFFSET mstr+1   ; convert minutes
   mov cx,2
   mov al,mins
   call bin2dec
 
   mov bx, OFFSET sstr+1   ; convert seconds
   mov cx,2
   mov al,secs
   call bin2dec
 
   mov bx, OFFSET dstr+1   ; convert hundreths
   mov cx,2
   mov al,hund
   call bin2dec
   ret
;----------------------------------------------------------------------
; bin2dec converts a 16-bit unsigned integer into ASCII decimal 
; Input Parameters:
; AX: number to convert
; BX: pointer to least significant digit of output buffer
; CX: number of bytes desired in output buffer (1-5)
; Returns: 
; Buffer filled with digits
; Destroys contents of AX, BX, CX, DX and SI

Bin2dec:
mov si, 10 ; divisor for DIV instruction

L1:
sub dx, dx ; since a 16 bit divides DX:AX by the divisor

; we need to zero-extend AX into DX:AX
div si ; divide by 10; remainder in DX and quotient

; is in AX. Remainder is next least sig. digit
add dl, '0' ; convert binary to ASCII (same as add dl, 30h)
mov [bx], dl ; store in output buffer 
dec bx ; point to next char in output 
loop L1 ; repeat for CX chars
ret



; DISPTIME.A86 -- Version 2
; Obtain system time from DOS and display it on the console
 
 jmp start
   msg  DB 'The time is '
   hstr DB 2 DUP ?,':'   ; string to store hours,
   mstr DB 2 DUP ?,':'   ; minutes
   sstr DB 2 DUP ?,'.'   ; seconds 
   dstr DB 2 DUP ?,'$'   ; and hundredths of seconds
 
   hrs  DB ?             ; binary storage for results
   mins DB ?             ; of DOS Get Time function
   secs DB ?
   hund DB ?
 
 start:
 mov ah,2CH               ; DOS Get System Time function code
 int 21H
 mov hrs,ch               ; store the results
 mov mins,cl
 mov secs,dh
 mov hund,dl
 
 call timestr             ; convert time to a string
 mov dx,OFFSET msg        ; display on the console
 mov ah,09H               ; function code for Display String
 int 21H
 mov ax,4C00H             ; DOS terminate program
 int 21H

 timestr:
   mov cx,4            ; we want to convert 4 numbers
   mov dl,10           ; used to divide by 10
   mov bx, OFFSET hstr ; bx points to string buffer 
                       ; for ASCII chars
   mov si, OFFSET hrs  ; and si points to number to convert
 L1:
   mov al,[si]         ; load number to convert
   sub ah,ah           ; zero out AH because we're 
                       ; dividing a 2 digit number
   div dl              ; divide AX by 10. Quotient in AL,
                       ; remainder in AH
   or ax,3030H         ; convert AX to ASCII digits
   mov [bx],ax         ; store result. Bytes are swapped 
                       ; to correct order
   add bx,3            ; point to next string
   inc si              ; and next number
   loop L1             ; and do it
   ret

